
Hoofdstuk 10.

Meer over recursieve algoritmen.

Recursie laat toe een groot aantal algoritmen op een eenvoudige en duidelijke wijze uit te drukken.

In sommige gevallen echter kan recusieve programmatie uiterst inefficiënt zijn omdat het kan resulteren in een ongeoorloofd groot aantal procedure activaties. Telkens een recursief algoritme geïmplementeerd wordt, dient men aandachtig de meerkost in uitvoeringstijd en in geheugenruimte van de procedureoproepen af te wegen tegenover de elegantie van de programmatie.

10.1. Pro’s en Contra’s van recursieve algoritmen.

10.1.1. Voor de hand liggende recursie.

Wiskundige problemen met een gehele onafhankelijke variabele n kunnen zeer vaak met de inductie methode opgelost worden:

- Eerst veronderstelt men dat een oplossing voor een bepaalde waarde van n gekend is.

- dan wordt er een “volgende stap” uitgewerkt voor (n+1), die gebruik maakt van de oplossing voor n.

- tenslotte tracht men een bijzondere waarde n0 te vinden voor dewelke de oplossing triviaal is.

Hiermee kan het probleem voor elke waarde van m>n0 door opeenvolgende toepassing van de “volgende stap” opgelost worden.

De “volgende stap” implementeren onder de vorm van een recursieve procedure met n als parameter leidt meestal tot zeer duidelijke programma's. Dergelijke procedure moet uiteraard een test op de waarde van n bevatten om de triviale oplossing voor n0 te kunnen invoeren en zo de recursie te beperken.

Een eerste zeer eenvoudige illustratie hiervan vindt men in het berekenen van n!. In tabel 10.1 vindt men het algoritme en de overeenstemmende procedure.

	
	The factorial function

	Step-up function:
	Fac(n+1) = (n+1) * Fac(n)

	Trivial solution:
	Fac(1) = 1

	Recursive procedure:
	PROCEDURE Fac(n:CARDINAL):CARDINAL;

BEGIN
  IF n > 1 

    THEN RETURN n * Fac(n-1)

    ELSE RETURN 1

  END (* IF *)

END Fac;


Table 10.1. De recursieve faculteitsfunctie.

Een ietwat complexer maar erg gelijkaardig voorbeeld treft men aan bij het berekenen van de Fibonaci getallen. Tabel 10.2 geeft de definitie van die getallen en een recursieve procedure om ze te berekenen.

	
	The Fibonaci function

	Step-up function:
	Fib(n+1) = Fib(n) + Fib(n-1)

	Trivial solution:
	Fib(1) = 1; Fib(0) = 0

	Recursive procedure:
	PROCEDURE Fib(n:CARDINAL):CARDINAL;

BEGIN
  IF n > 1 

    THEN RETURN Fib(n-1)+Fib(n-2)

    ELSIF n = 1 THEN RETURN 1

    ELSE             RETURN 0

  END (* IF *)

END Fib;


Tabel 10.2. De recursieve Fibonaci functie.

Het “Torens van Hanoi” probleem besproken in hoofdstuk 7 werd op exact dezelfde wijze benaderd: men veronderstelde dat het probleem kon opgelost worden voor n ringen, dan werd de “volgende stap” voor (n+1) ringen ontwikkeld en uiteindelijk werd de triviale oplossing voor n=0 ingevoerd. Tabel 10.3 beschrijft het algoritme en het overeenstemmend programma. 

	
	Moving one tower with n rings from A to B using C

	Step-up function:
	Move the n-1 upper rings from A to C

Move one ring from A to B

Move the n-1 rings from C to B

	Trivial solution:
	Moving a tower with no rings

	
	PROCEDURE MoveTower(Height:CARDINAL;

                   From,Towards,Using:CHAR);


  PROCEDURE MoveDisk(TakeOff,PutOn:CHAR);
  BEGIN
  ..
  END MoveDisk;


BEGIN
  IF Height > 0 THEN
    MoveTower(Height-1,From,Using,Towards);
    MoveDisk(From,Towards);
    MoveTower(Height-1,Using,Towards,From)
  END
END MoveTower;


Table 10.3. Simulatie van het “torens van Hanoi” verhaal.

10.1.2. Performantie van bepaalde recursieve algoritmen

10.1.2.1. Fakulteit.

In dit geval wordt de procedure Fac n maal opgeroepen. Daar deze procedure alleen maar een IF test en een vermenigvuldiging uitvoert is het duidelijk dat het oproepen van de procedure relatief veel duurder (zowel qua tijd als qua geheugenruimte) zal zijn dan het verrichte werk. Tabel 10.4 toont een alternatieve procedure Fac die net hetzelfde nuttig werk verricht, maar zonder de meerprijs van de procedure activaties.

	PROCEDURE Fac(n:CARDINAL):CARDINAL;

  VAR f:CARDINAL;

BEGIN
  f := 1;

  WHILE n > 1 DO f := f * n; n := n-1 END;

  RETURN f

END Fac;


Table 10.4. A non-recursive factorial procedure.

10.1.2.2. Fibonaci.

In het geval van de Fibonaci procedure is de prijs van recursie veel hoger vermits de procedure, voor alle waarden van n groter dan 1 zichzelf twee maal moet oproepen. Dit betekent dat het aantal oproepen exponentieel zal stijgen met n, wat duidelijk onwerkbaar wordt voor ietwat grote waarden van n.

Tabel 10.5 toont een alternatieve versie van de procedure Fib waaruit de onaanvaardbare meerkost van recursiviteit geëlimineerd is. Deze procedure verstaan is misschien wel moeilijker dan voor de recursieve versie, maar de snelheid is spectaculair hoger. 

	PROCEDURE Fib (n:CARDINAL): CARDINAL;

VAR
  i, fn, fnm1, fnm2 : CARDINAL;

BEGIN
  IF n=0 THEN RETURN 0

  ELSIF n=1 THEN RETURN 1

  ELSE
    fnm2 := 0;

    fnm1 := 1;

    FOR i := 2 TO n DO
      fn := fnm1 + fnm2;

      fnm2 := fnm1;

      fnm1 := fn

    END;

    RETURN fn

  END;

END Fibonacci;


Table 10.5. Niet-recursieve Fibonaci procedure.

10.1.2.3. Torens van Hanoi.

De situatie is gans anders met de torens van Hanoi: de computer wordt hier niet gebruikt om een bepaalde oplossing te vinden maar om de uitvoering, door imaginaire priesters, van een welbepaald algoritme te simuleren. De uitvoeringssnelheid van het eigenlijke algoritme is quasi irrelevant daar het grootste deel van de tijd zal gespendeerd worden aan het voorstellen van de bewegingen van de ringen op een scherm. Niettemin kan men opmerken dat het “Torens van Hanoi” algoritme ook in de orde van 2n procedure oproepen vergt om een toren met n ringen te verplaatsen, maar dat het schrijven van een niet recursieve procedure voor een willekeurig aantal ringen een niet triviale opdracht is.

10.1.2.4. Binair zoeken.

In dit algoritme, besproken in hoofdstuk 8, is het aantal oproepen van de zoekprocedure bijzonder klein, vermits zoeken in een tabel met n elementen maar log2n oproepen vergt.

Niettemin, daar het werk dat de procedure zelf verricht bijzonder eenvoudig is, werd er in hoofdstuk 8 een alternatieve vorm van die procedure voorgesteld, die net hetzelfde doet maar zonder recursieve procedure oproepen. Deze alternatieve versie bracht een meetbare performantieverhoging mee, maar vermits het binair zoekalgoritme toch al bijzonder snel is, is de tijdwinst zelden belangrijk.

10.1.2.5. Quicksort.

Het quicksort algoritme dat in hoofdstuk 9 besproken werd is bijzonder goed om een grote tabel op te splitsten in een geordende verzameling aaneensluitende kleine tabellen die elk nog moeten geordend worden. Gebruik maken van Quicksort om deze kleine tabellen te ordenen zou aanleiding geven tot een groot aantal (een veelvoud van het aantal elementen in de tabel) recursieve oproepen. In hoofdstuk 9 werd een techniek besproken om dit groot aantal oproepen te beperken door beroep te doen op “insertion sort” voor het afwerken.

10.1.3. Besluiten.

Uit het voorgaande is het vrij duidelijk dat recursieve programmatie dient vermeden te worden wanneer recursie zou resulteren in een groot aantal activaties van een eenvoudige procedure.

Daartegenover staat dat recursie de aangewezen strategie is wanneer het leidt tot een complexe procedure die een klein aantal keren moet geactiveerd worden. Deze strategie zal inderdaad leiden tot eenvoudig te schrijven en te verifiëren programma's waarvan de performantie maar weinig zal beinvloed worden door de procedure activaties.

Tenslotte, wanneer een complexe procedure een groot aantal keren moet geactiveerd worden, zal de programmeur erg zorgvuldig de prijs van een groot aantal procedure activaties moeten afwegen tegen het comfort van eenvoudige en duidelijke algoritmen.

Teneinde het comfort van recursieve algoritmen te combineren met de efficiëntie van niet recursieve algoritmen, worden er voor bepaalde programmeertalen waarin recursie centraal staat, compilers ontwikkeld die, in de mate van het mogelijke, recursieve procedures vervangen door iteratieve object code.

10.2 Backtracking.

10.2.1. Inleiding.

Wanneer, om een bepaald probleem op te lossen, meerdere keuzes achtereenvolgens moeten gemaakt worden, zonder dat het mogelijk is de gevolgen van de eerste keuzes op de latere volledig in te schatten, moet men zgn. “backtracking” technieken gebruiken. Deze oplossingstechniek bestaat erin systematisch alle mogelijke combinaties van keuzes te exploreren. De naam “backtracking” alludeert aan het terug komen op een vroegere keuze wanneer blijkt dat geen van de keuzes in de volgende stap naar een oplossing kan leiden.

Deze algoritmen zijn vrij belangrijk in de ingenieurstechnieken en kunnen elegant opgelost worden door middel van recursieve procedures, zonder dat de recursiviteit de efficiëntie noemenswaardig aantast.

Het intuïtief meest eenvoudige probleem waarvoor “backtraking” technieken nodig zijn is het doorlopen van een doolhof zonder lussen. Dergelijke doolhof wordt schematisch voorgesteld in fig. 10.1. Sommige knooppunten hebben meerdere uitgangen, die naar volgende knooppunten leiden, terwijl andere geen enkele uitgang hebben. Sommige knoopunten hebben een uitgang die naar de buitenwereld leidt. Het is de bedoeling in de doolhof te treden via het vertrek knooppunt en dan een van de uitgangen (of alle uitgangen) te vinden.

Fig.10.1. Een doolhof zonder lussen met een ingang en twee uitgangen.

Om een weg door deze doolhof te vinden met behulp van een backtracking algoritme gebruikt men een recursieve procedure die, vanuit een welbepaald knooppunt, alle uitgangen systematisch probeert. Een uitgang proberen bestaat erin voor deze procedure, zichzelf opnieuw op te roepen, maar nu voor het knoopunt dat via de uitgeprobeerde uitgang bereikt werd. Wanneer in een knoopunt er geen ongeprobeerde uitgangen overblijven, keert de procedure gewoonweg terug naar haar oproeper, die dan een volgende uitgang van het vorige knooppunt kan gaan exploreren. Indien een uitgang naar de buitenwereld gevonden is, wordt de overeenstemmende weg genoteerd en wordt het algoritme beëindigd. Indien men alle oplossingen zoekt, gaat het backtracking algoritme gewoon verder nadat een oplossing genoteerd is geweest.

Het aantal recursieve procedure oproepen nodig om een probleem met backtracking op te lossen is maximaal gelijk aan het aantal alternatieven in elk knoopunt, verheven tot de diepte van de recursie (het aantal versies van de procedure die simultaan actief kunnen zijn) is gelijk aan het aantal knoopunten dat moet doorlopen worden om de oplossing te bereiken. In het voorbeeld van fig.10.1 bedraagt het aantal procedure-oproepen maximaal 33 (=81) en is de recursiediepte maximaal 3.

Backtracking algoritmen kunnen onaanvaardbaar traag worden wanneer het aantal alternatieven per knooppunt of de recursiediepte overdreven groot worden. In praktijk, wanneer een van die parameters enkele tientallen overschrijdt, zijn de eenvoudige backtraking algoritmen niet meer realistisch bruikbaar.

10.2.2. Het “Acht Koninginnen” probleem.

Vaak illustreert men de mogelijkheden van backtracking aan de hand van voorbeelden uit de schaakwereld omdat zij wel omschreven zijn en weinig contextinformatie vergen. Het “acht koninginnen probleem” is een dergelijk voorbeeld: men wenst op een 8 * 8 schaakbord acht koninginnen te plaatsen zonder dat ze elkander zouden bedreigen. Terwijl het praktisch belang van dit probleem eerder beperkt is, is het algoritme niettemin zeer belangrijk want het acht koninginnen probleem is representatief voor vele praktische problemen in het domein van het computer ondersteund ontwerpen zoals bv. het plaatsen van doorverbindingen op een gedrukte elektronische schakeling.

Het schaakbord wordt voorgesteld door middel van een  ARRAY[8,8] OF BOOLEAN, waarbij een vrije plaats voorgesteld wordt door de waarde TRUE en een konigin door de waarde FALSE. volgens de regels van het schaakspel kan een koningin horizontaal, verticaal en volgens de diagonalen bewegen. Hieruit volgt op evidente wijze dat er een konigin per rij zal moeten komen. Er zal dus een procedure Try(Rij) geschreven worden die zal pogen een veilige plaats in die rij te vinden. Wanneer dergelijke veilige plaats gevonden is zal deze procedure zichzelf recursief oproepen teneinde een koningin op de volgende rij te plaatsen. Fig.10.3 toont het NS diagramma van dergelijke procedure.

Fig.10.3. Try procedure voor het acht koniginnen probleem.

Om na te gaan of een bepaalde positie veilig is voor een koningin wordt de functieprocedure Safe(Row,Col) gebruikt. Wanneer deze procedure opgeroepen wordt gaat ze controleren of er in geen van de voorgaande rijen een koningin staat langs de verticale of langs een van de twee diagonalen. Teneinde deze procedure zo eenvoudig mogelijk te houden werd het schaakbord langs beide zijden uitgebreid zodanig dat, vanuit elk willekeurig punt van het oorspronkelijke schaakbord beide diagonalen volledig kunnen doorlopen worden zonder het uitgebreide bord te verlaten. Deze techniek verzekert zeker niet de maximale snelheid voor de procedure Safe, maar werd gekozen voor haar eenvoud.

Tabel 10.6 bevat de voornaamste delen van een programma dat het “acht konniginnen probleem” oplost. Het vindt 92 verschillende oplossingen (in feite, wanneer symmetrieën beschouwd worden, ziet men dat er 23 totaal verschillende oplossingen bestaan)

	MODULE Queens;

...

  VAR  Board : ARRAY[-6..15],[1..8] OF BOOLEAN;

       Col : [-6..15]; Row : [1..8];

  PROCEDURE PrintSolution;

  BEGIN

    ...
  END PrintSolution;

  PROCEDURE Try(Row : CARDINAL);

    VAR Col : CARDINAL;

    PROCEDURE Safe(Col,Row:INTEGER):BOOLEAN;

      VAR r : INTEGER; f : BOOLEAN;

    BEGIN
      f := TRUE;

      FOR r := 1 TO Row-1 DO
          f := f AND Board[Col,r] 

          AND Board[Col+Row-r,r] AND Board[Col-Row+r,r]

      END;

      RETURN f

    END Safe;

  BEGIN
    FOR Col := 1 TO 8 DO

      IF Safe(Col,Row) THEN
        Board[Col,Row] := FALSE;

        IF Row < 8 THEN Try(Row + 1)

                   ELSE PrintSolution

        END;
        Board[Col,Row] := TRUE;

      END;
    END
  END Try;

BEGIN
  (* Board Initialisation *)

  FOR Col := -6 TO 15 DO

    FOR Row := 1 TO 8 DO

      Board[Col,Row] := TRUE

    END
  END;
  (* Backtracking algorithm *)

  Try(1);

END Queens.


Table 10.6. A recursive solution for the eight queens problem.

Vermits een koningin op elke rij acht verschillende posities zou kunnen innemen en er acht rijen zijn is een eerste voor de hand liggende bovenlimiet voor het aantal oproepen van Try gelijk aan 87(( 2.106). Rekening houden in de berekening van deze bovenlimiet met de koninginnen die op een vertikale staan vervangt de bovenlimiet van 87 door 8! wat al veel kleiner is.

Om meer inzicht te krijgen in de voor- en nadelen van recursieve algoritmen kan men expliciet de bewerkingen die door elke versie van Try uitgevoerd worden uitschrijven. Elke versie van Try zoekt een plaats voor een koningin in een bepaalde rij en start dan een gelijkaardige zoekoperatie in de volgende rij. Expliciete programmatie hiervan resulteert in 8 geneste FOR lussen, een per rij, zoals zichtbaar in tabel 10.7.

Vergelijking tussen de recursieve en niet recursieve versies van het backtracking algoritme toont duidelijk dat de recursieve versie eenvoudiger en eleganter is dan de niet recursieve. In het geval van het acht koninginnen probleem heeft deze elegantie geen effect op de performantie (110 mS voor beide versies op een Compaq Contura 400CX). Dit is niet verrassend wanneer men vaststelt dat ongeveer 90% van de tijd gespendeerd wordt in de procedure Safe die uiteraard dezelfde is voor beide versies.

10.2.3. De wandeling van het paard.

Een paard, vertrekkend uit de hoek van een schaakbord wenst alle vakjes van het bord juist een maal te bezoeken. De oplossing kan gevonden worden met behulp van een recursieve procedure die, vanuit een bepaald vakje alle bereikbare plaatsen afgaat en, in elk van deze plaatsen zichzelf terug oproept.

Het schaakbord wordt voorgesteld door middel van een ARRAY[1..8,1..8] OF CARDINAL. De array wordt initiëel opgevuld met nullen en telkens het paard een bepaald vakje bereikt, wordt het sequentiëel genummerd. Enkel sprongen naar een vakje dat een nul bevat wordt uitgevoerd. Bewegingen die buiten het schaakbord zouden terecht komen, kunnen opgevangen worden door het schaakbord langs alle kanten uit te breiden met met twee rijen, geinitializeerd met een waarde verschillend van nul.

De implementatie van dit algoritme wordt als oefening voorgesteld aan de studenten. Om hun programma te testen doen zij er echter goed aan te werken met een gereduceerd schaakbord (6*6 is een goede maat) want de volledige oplossing op een 8*8 schaakbord vergt in de orde van 863 stappen, wat op een hedendaagse computer ongeveer 30 miljoen jaren zou kunnen duren.

	MODULE Queens;

...

  VAR  Board : ARRAY[-6..15],[1..8] OF BOOLEAN;

       Col : [-6..15]; Row : [1..8];

       Col1,Col2,Col3,Col4,Col5,Col6,Col7,Col8: CARDINAL;

  PROCEDURE PrintSolution;

  BEGIN

    ...
  END PrintSolution;

  PROCEDURE Safe(Col,Row:INTEGER):BOOLEAN;

    (* unchanged *)

  END Safe;

BEGIN
  (* Board Initialisation *)

  FOR Col := -6 TO 15 DO

    FOR Row := 1 TO 8 DO

      Board[Col,Row] := TRUE

    END
  END;
  (* Backtracking algorithm *)

  FOR Col1 := 1 TO 8 DO

   IF Safe(Col1,1) THEN
    Board[Col1,1] := FALSE;

    FOR Col2 := 1 TO 8 DO

     IF Safe(Col2,2) THEN
      Board[Col2,2] := FALSE;

      ...

        FOR Col8 := 1 TO 8 DO

         IF Safe(Col8,8) THEN
          Board[Col8,8] := FALSE;

          PrintSolution;

          Board[Col8,8] := TRUE;

         END;

        END; (* FOR 8 *)

      ...

      Board[Col2,2] := TRUE;

     END; (* IF 2 *)

    END; (* FOR 2 *)

    Board[Col1,1] := TRUE;

   END; (* IF 1 *)

  END; (* FOR 1 *)

END Queens.


Table 10.7. Non recursive backtracking for the eight queens problem.

10.2.4. Ontwerp van een multi-frekwentie generator.

Om te tonen hoe het acht koninginnen algoritme kan aangewend worden om praktische ingenieursproblemen op te lossen zal hier de synthese van een frekwentie generator voor een telecommunicatie toepassing beschreven worden.

Zes verschillende frekwenties, 875 Hz, 1025 Hz, 1475 Hz, 1625 Hz, 1925 Hz en 2075 Hz moeten met behulp van (gehele) frekwentie delers afgeleid worden van een enkele meester oscillator. (fig 10.4). 

Fig. 10.4. Block diagram of the frequency synthesizer.

De opgewekte frekwenties mogen niet meer dan 1 Hz afwijken van de nominale frekwenties en de meester oscilator moet een frekwentie hebben die tussen 1 en 10 MHz ligt.

De voor de hand liggende oplossing zou er in bestaan het kleinste gemeenschappelijk veelvoud van de zes frekwenties te bepalen. Jammer genoeg ligt dat kleinste gemeenschappelijk veelvoud (25,122,222,125 Hz) buiten de toegelaten frekwentieband voor de meester oscillator.

De factoren van de zes frekwentie delers kunnen als zes onafhankelijke variabelen beschouwd worden waarvan alle mogelijke combinaties dienen onderzocht te worden, net zoals de posities van de koninginnen op elke rij onderzocht werden. In plaats van te variëren tussen 1 en 8 zullen deze variabelen moeten variëren in een gamma dat bepaald wordt door de toelaatbare frekwentie van de meester oscillator. Tengevolge de tolerantie van +/- 1 Hz op elk van de gevraagde frekwenties zal elke factor een band van aanvaardbare frekwenties voor de meester oscillator bepalen. Een verzameling van 6 factoren voor dewelke de aanvaardbare frekwentiebanden een gemeenschappelijk deel bezitten geeft een aanvaardbare oplossing voor het gestelde probleem. Een verdere optimizatie zou er in kunnen bestaan alle verzamelingen van 6 factoren die tot een oplossing leiden te bepalen en dan uit die verzameling deze te kiezen voor dewelke de totale fout ten opzichte van de nominale frekwenties minimaal is. Fig.10.6 toont het NS diagramma van de Try procedure voor dit probleem. De parameter Fr is de index die bepaalt welke van de 6 frekwenties behandeld wordt in een specifieke oproep van de procedure Try.

Tabel 3.8 toont een programma dat met behulp van backtracking alle mogelijke verzamelingen van 6 factoren en de overeenstemmende frekwenties van de meester oscillator bepaalt

Een modem waarvan de frekwentiegenerator met het hier beschreven programma werd ontworpen is gedurende meer dan tien jaar geproduceerd geweest door een belgisch bedrijf, en wordt nog steeds gebruikt op veel schepen om weerberichten te ontvangen.

Fig.10.5. Recursive Try procedure for the frequency synthesizer.

10.3. Optimizatie algoritmen met backtracking.

Backtracking kan ook nog gebruikt worden in bepaalde optimizatie problemen : voor alle mogelijke oplossingen wordt er een kostfunctie geëvalueerd. Tijdens het zoeken naar oplossingen wordt de kleinste waarde van de kostfunctie die reeds gevonden is continu bijgehouden en wordt het opbouwen van een nieuwe oplossing onderbroken van zodra blijkt dat de uiteindelijke kost groter zal zijn dan het reeds gevonden minimum.

10.3.1. Het handelsreizigers probleem.

Het handelsreizigersprobleem is een klassiek voorbeeld van dergelijke optimizatie: een handelsreiziger moet een zeker aantal plaatsen bezoeken en wenst zo weinig mogelijk kilometers af te leggen. Elke weg die langs alle plaatsen passeert is een mogelijke oplossing, maar men zoekt de kortste van deze oplossingen. De backtracking oplossing bestaat erin een procedure te schrijven die, in een bepaalde plaats, de reeds afgelegde afstand bepaalt en, indien deze afstand niet groter is dan de kortste reeds gevonden oplossing, systematisch alle overblijvende bestemmingen uitprobeert, telkens door zichzelf recursief op te roepen.

Wanneer in een bepaalde plaats de minimale afstand overschreden wordt, gaat men terug naar de vorige plaats en loopt men daar verder alle mogelijke bestemmingen af.

Wanneer het einddoel bereikt wordt, met een totale afstand die kleiner is dan de kortst gekende afstand, wordt deze laatste aangepast en wordt de gevonden weg geregistreerd.

Met dergelijke optimizatie algoritmen kan men trachten een reeds gekende oplossing te verbeteren, zelfs wanneer een volledige oplossing prohibitief veel rekentijd in beslag zou nemen. Om met dergelijke werkwijze snel succes te boeken is het wel belangrijjk dat de eerste oplossing reeds vrij goed is om het opbouwen van minder goede alternatieven zo vlug mogelijk te kunnen stoppen. Daarbij loont het dan ook nog sterk de moeite vanuit een bepaalde plaats de overblijvende bestemmingen in een optimale volgorde uit te proberen. Zo is het experimenteel gekend dat de dichtsbij gelegen bestemmingen eerst uitproberen snel naar goede oplossingen kan leiden. De kunst van het bepalen in welke volgorde de verschillende alternatieven moet geëxploreerd worden bij backtracking algoritmen vormt een belangrijk studiegebied in de tak van de informatica die men “artificiële intelligentie” noemt.

Fig.10.6 The recursive backtracking procedure for the traveling salesman problem.

Wanneer echter het aantal opties die moeten geselecteerd worden enkele tientallen overschrijdt worden backtracking algoritmen onmogelijk traag. 

Een alternatieve benadering die dan kan geprobeerd worden is geïnspireerd door de evolutieleer: de waarden van deelverzamelingen variabelen van verschillende goede oplossingen worden op toevalligere wijze gecombineerd in nieuwe oplossingen. Indien deze beter zijn dan de “ouders” zullen zij verder mee gebruikt worden om naar nog betere oplossingen te evolueren.

10.4. Fractalen.

Fractalen zijn geometrische figuren waarin een zelfde patroon, op verschillende schalen, herhaaldelijk voorkomt. Zij kunnen kunnen voor decoratieve doeleinden gebruikt worden maar kunnen ook aangewend worden, onder de vorm van printplaten als breedbandige antenne voor bv. multi-band draagbare telefoons. Programma's om dergelijke figuren te tekenen kunnen ook best als recursieve procedures gestructureerd worden.

Indien we aannemen dat de procedure RECTANGLE(a,b,c,d) een rechthoek tekent waarvan de linker onderhoek en de rechter bovenhoek respectievelijk de coordinaten (a,b) en (c,d) hebben, dan tekent het programma weergegeven in fig. 10.7 de fractale tekening weergegeven in fig.10.8. Dergelijke figuur noemt men een Serpinsky curve met een recursieve diepte van 3.

	PROCEDURE FractalBox(x,y,d,n:CARDINAL);
  PROCEDURE Box(x,y,d:CARDINAL,color);

  BEGIN

    Rectangle(x-d,y-d,x+d,y+d,color)

  END Box;
BEGIN

  Rectangle(x-d,y-d,x+d,y+d);

  n := n-1

  IF n > 0 THEN

    FractalBox(x-d,y-d,d DIV 2,n);

    FractalBox(x-d,y+d,d DIV 2,n);

    FractalBox(x+d,y-d,d DIV 2,n);

    FractalBox(x+d,y+d,d DIV 2,n);

  END

END FractalBox;


Fig.10.7. Procedure om Serpinsky curves met een diepte n te tekenen.

[image: image6.wmf]Fig.10.8. Serpinsky Curve met diepte 3.

